# Maintainer: Chih-Hsuan Yen <yan12125@archlinux.org>
# Contributor: Jerome Leclanche <jerome@leclan.ch>

pkgname=lxqt-config
pkgver=1.3.0
pkgrel=1
pkgdesc="LXQt system configuration."
arch=("loong64" "x86_64")
groups=("lxqt")
url="https://github.com/lxqt/$pkgname"
license=("LGPL2.1")
depends=("glibc" "gcc-libs"
         "qt5-base" "qt5-svg" "qt5-x11extras" "libkscreen"
         "libQt5Xdg.so" "liblxqt" "lxqt-themes"
         "libx11" "libxcb" "libxcursor" "libxfixes" "libxi" "xf86-input-libinput"
         "libudev.so" "zlib")
makedepends=("lxqt-build-tools")
optdepends=(
	'xsettingsd: help GTK applications apply themes on the fly'
)
source=(
	"https://github.com/lxqt/$pkgname/releases/download/$pkgver/$pkgname-$pkgver.tar.xz"
	"https://github.com/lxqt/$pkgname/releases/download/$pkgver/$pkgname-$pkgver.tar.xz.asc"
)
sha256sums=('929f60b58cf8dbf84f0f778a59c9794bfcc74abb8ea68177e28deb666eae698f'
            'SKIP')
validpgpkeys=(
	"19DFDF3A579BD509DBB572D8BE793007AD22DF7E"  # https://github.com/tsujan.gpg
)

build() {
	mkdir -p build
	cd build
	cmake "$srcdir/$pkgname-$pkgver" \
		-DCMAKE_INSTALL_PREFIX=/usr \
		-DCMAKE_BUILD_TYPE=None
	make
}

package() {
	cd build
	make DESTDIR="$pkgdir" install
}
